```julia
using LinearAlgebra, Plots
```

It's usually quite straightforward how to make sense of the vector p-norms:

$$\|x\|_1 = \sum_{i=1}^n |x_i|, \qquad \|x\|_2 = \sqrt{\sum_{i=1}^n |x_i|^2}, \qquad \|x\|_\infty = \max_{1\le i\le n} |x_i|.$$

What is perhaps not so obvious are induced matrix norms:

$$\|A\|_p = \sup_{0\ne x\in\mathbb{C}^n} \frac{\|Ax\|_p}{\|x\|_p} = \max_{\|x\|_p\le 1} \|Ax\|_p = \max_{\|x\|_p=1} \|Ax\|_p.$$

To visualize these, we will restrict our attention to $m = n = 2$; that is, vectors live in the plane. For $p = 1$ and $p = \infty$, we get lozenges, and for $p = 2$, we get ellipses.

```julia
@bind θ html"""<input type="range" min=0 max=2 step=0.01 value=0>"""
```

```julia
@bind p html"""<input type="range" min=1 max=10 step=0.1 value=1>"""
```

```
(0,  1)
```

```julia
θ, p
```

```
A = 2×2 Array{Float64,2}:
     1.0  2.0
     0.0  2.0
```

```julia
A = [1 2; 0 2.0]
```




```julia
begin
    t = 0:0.01:2
    v = [normalize!([cospi(t), sinpi(t)], p) for t in t]
    u = [A*v for v in v]
    plot(map(u->u[1], u), map(u->u[2], u); label="Map of the unit circle in the $p-norm.", legend = :topleft)
    x = normalize!([cospi(θ), sinpi(θ)], p)
    y = A*x
    plot!([0,y[1]], [0,y[2]]; label="Vector with angle $θ*π.")
    xlims!(-3,3)
    ylims!(-3,3)
end
```

For matrices, the induced p-norms are accessed in Julia via `opnorm`, short for operator norm (because a matrix is an archetypal linear operator).

```
(4.0, 2.92081, 3.0)
```

```julia
opnorm(A, 1), opnorm(A, 2), opnorm(A, Inf)
```

Similarly, it is important to grasp a strong intuition on p-norms of functions. This is most easily visualized when looking at an approximation error. Suppose we wish to approximation $f(x) = e^x$ on $[-1,1]$. Then a degree-2 polynomial interpolant through the points $\{-1,0,1\}$ has the form:

$$p_2(x) = e^{-1}\frac{x(x-1)}{2} - (x-1)(x+1) + e^1\frac{x(x+1)}{2}.$$

Let's plot the integrand in the p-norm:

$$\|f - p_2\|_p = \left(\int_{-1}^{1} |f(x) - p_2(x)|^p \, \mathrm{d}x \right)^{\frac{1}{p}}.$$




```
begin
    s = range(-1, stop = 1, length=1001)
    f = x -> exp(x)
    p2 = x -> exp(-1)*x*(x-1)/2 - (x-1)*(x+1) + exp(1)*x*(x+1)/2
    l = @layout([a b])
    plot(
        plot(s, [f.(s), p2.(s)]; label=permutedims(["\$f(x)\$", "\$p_2(x)\$"]), legend=:topleft),
        plot(s, abs.(f.(s) .- p2.(s)).^p; label="\$|f(x)-p_2(x)|^{$(p)}\$", legend=:topleft)
    )
end
```